

JEUX EN BASIC SUR

TO 7



JEUX EN BASIC SUR

TO 7

Auteur : Pierre **MONSAUT**
Illustrations : Jean-François Penichoux
Photo Lucky Mômes

Tous les efforts ont été faits pour fournir dans ce livre une information complète et exacte. Néanmoins, SYBEX n'assume de responsabilités ni pour son utilisation, ni pour les contrefaçons de brevets ou atteintes aux droits de tierces personnes qui pourraient résulter de cette utilisation.

Copyright version originale © 1984, Sybex

ISBN 2-7361-0026-2

Tous droits réservés. Toute reproduction même partielle, par quelque procédé que ce soit, est interdite sans autorisation préalable. Une copie par xérographie, photographie, film, bande magnétique ou autre, constitue une contrefaçon passible des peines prévues par la loi sur la protection des droits d'auteur.

JEUX EN BASIC SUR

TO 7

PIERRE MONSAUT



Paris • Berkeley • Düsseldorf


Table des matières

Squash	9
Blitz	13
D.C.A.	17
Trace	21
Slalom	25
Ramasse-miettes	29
Tank	33
Alphabet	37
Atterrissage	41
Chasse au canard	45
Serpent	51
Simon	55
Grand-prix	59
Crabes	63
Casse-briques	69
Poursuite	75
Robots	81
Parachute	87

Avant-Propos

Les jeux présentés dans ce livre utilisent des fonctions propres à votre microordinateur : son, couleur, graphisme haute résolution, caractères graphiques définis par l'utilisateur, etc. Pour cette raison, ces programmes ne fonctionnent que sur le TO 7 et sont difficilement adaptables à des matériels différents.

Au-delà du jeu lui-même, l'étude de ces programmes vous enseignera de nombreuses techniques de programmation qui vous seront très utiles pour développer vos propres programmes.



NOTES CONCERNANT LES PROGRAMMES

Les programmes doivent être tapés tels qu'ils sont reproduits. Ils doivent être enregistrés sur cassette (ou sur disquette) pour éviter d'avoir à les retaper à chaque utilisation. Utilisez pour cela l'instruction

SAVE " < nom > "

en remplaçant < nom > par le nom du programme à enregistrer.

Exemple : pour enregistrer le programme BLITZ, tapez

SAVE "BLITZ"

et pressez la touche RETURN après avoir appuyé sur les touches PLAY et RECORD du magnétophone. Pour recharger un programme, rembobinez la cassette jusqu'au début du programme (certains magnétophones possèdent un compteur, très utile si vous avez pris soin de noter le nombre indiqué avant l'enregistrement) et tapez LOAD " < nom > ". Si vous voulez, par exemple, recharger le programme BLITZ, tapez


LOAD "BLITZ"

et pressez la touche RETURN.

Afin de réduire au maximum les risques d'erreurs, les listings des programmes ont été reproduits par photographie. Si vos programmes ne fonctionnent pas, vérifiez-les attentivement en considérant plus particulièrement les points suivants :

- lignes oubliées,
- numéros de lignes erronés,
- erreurs de frappe (attention en particulier à ne pas confondre la lettre O et le chiffre 0 ou la lettre I et le chiffre 1).

Tous les jeux présentés fonctionnent en BASIC cassette sauf PARACHUTE et ATTERRISSAGE qui nécessitent le BASIC disque.



SQUASH

Grâce à votre ordinateur, vous pouvez jouer au squash assis dans un fauteuil. La raquette se déplace à l'aide du joystick ou des touches < Q > , < S > et barre d'espace. Vous disposez de dix balles que vous devez garder en jeu le plus longtemps possible. Chaque balle renvoyée rapporte un point.

```
10 REM *****
20 REM * SQUASH *
30 REM *****
40 CLEAR ,,2
50 GOSUB 660
60 GOTO 180
70 D=2*((STICK(0)=7)-(STICK(0)=3))
80 IF D<>0 THEN DO=D
90 IF STICK(0)=0 THEN DO=0
100 RX=RX+DO
110 RETURN
120 D$=INKEY$
130 D=2*((D$="Q")-(D$="S"))
140 IF D<>0 THEN DO=D
150 IF D$=N$ THEN DO=0
160 RX=RX+DO
170 RETURN
180 LOCATE BX,BY
190 COLOR 3,6
```

```
200 PRINT N$;
210 BX=BX+DX
220 BY=BY+DY
230 LOCATE BX,BY
240 PRINT B$;
250 IF BY=22 AND ABS(BX-RX-3)>1 THEN 360
260 IF BY=22 THEN S=S+1:BEEP:DY=-DY
270 IF BY=1 THEN BEEP:DY=-DY
280 IF BX=2 OR BX=37 THEN BEEP:DX=-DX
290 ON JS GOSUB 70,120
300 IF RX<0 THEN RX=0
310 IF RX>33 THEN RX=33
320 LOCATE RX,RY
330 COLOR 2
340 PRINT R$;
350 GOTO 180
360 NB=NB+1
370 IF NB=11 THEN 480
380 LOCATE BX,BY
390 PRINT N$;
400 FOR I=1 TO 3
410 BEEP
420 FOR J=1 TO 100
430 NEXT J
440 NEXT I
450 DO=0
460 GOSUB 940
470 GOTO 180
480 LOCATE 13,5
490 COLOR 0
500 PRINT "SCORE :";S;
510 IF S>R1 THEN R1=S
520 LOCATE 13,10
530 PRINT "RECORD :";R1;
540 LOCATE 13,15
550 PRINT "UNE AUTRE ?";
560 NB=0
570 S=0
580 D$=INKEY$
590 IF D$<>" " THEN 580
600 D$=INKEY$
610 IF D$=" " THEN 600
620 IF D$<>"N" THEN 50
```

```
630 SCREEN 4,6,6
640 CLS
650 END
660 CLS
670 SCREEN 1,6,6
680 ATTRB 1,1
690 DEFINT A-Z
700 DEFGR$(0)=255,255,255,0,0,0,0,0
710 DEFGR$(1)=24,126,126,255,255,126,126
,24
720 N$=CHR$(32)
730 LOCATE 1,10,0
740 PRINT "JOYSTICK (O ou N) ?";
750 D$=INKEY$
760 IF D$="" THEN 750
770 IF D$="O" THEN JS=1 ELSE JS=2
780 CLS
790 ATTRB 0,0
800 COLOR 6,1
810 FOR BX=1 TO 38
820 LOCATE BX,0
830 PRINT N$;
840 NEXT BX
850 FOR BY=1 TO 22
860 LOCATE 1,BY
870 PRINT N$;
880 LOCATE 38,BY
890 PRINT N$;
900 NEXT BY
910 R$=N$+N$+GR$(0)+GR$(0)+GR$(0)+N$+N$
920 B$=GR$(1)
930 RX=16
940 RY=23
950 BY=22
960 BX=INT(RND*34)+3
970 DY=-1
980 DX=(INT(RND*2)-0.5)*2
990 RETURN
```




BLITZ

Votre mission est de détruire la ville que vous survolez afin de pouvoir atterrir. A chaque passage, votre avion vole un peu plus bas. Vous ne pouvez larguer une bombe (en appuyant sur une touche quelconque) que lorsque la bombe précédente a atteint son objectif ou le sol. Lorsque votre avion a atterri (ou lorsqu'il s'est écrasé contre un immeuble), le score est affiché ainsi que le record du jour. Si ce jeu vous paraît trop difficile, vous pouvez changer les limites de la ville (10 et 30, ligne 860).

```
10 REM *****
20 REM * BLITZ *
30 REM *****
40 CLEAR ,,4
50 GOSUB 640
60 GOSUB 770
70 FOR HY=0 TO 23
80 FOR HX=0 TO 39
90 LOCATE HX, HY
100 Z=-Z
110 W=Z+0.5
120 COLOR 2,0
130 PRINT A$(W);
140 COLOR 1,0
150 IF HX>36 THEN 170
160 IF SCREEN (HX+3, HY)<>CN THEN 270
170 IF INKEY$<>" " AND BY=0 AND HX<>39 TH
EN BX=HX+1:BY=HY+1
180 IF BY<>0 THEN GOSUB 550 ELSE GOSUB 6
10
```

```
190 NEXT HX
200 LOCATE 38,HY
210 PRINT NN$;
220 NEXT HY
230 FOR I=1 TO 500
240 NEXT I
250 S=S+40*HY+HX
260 GOTO 60
270 S=S+40*HY+HX
280 LOCATE BX,B1
290 PRINT N$;
300 FOR I=HY TO 22
310 LOCATE HX+1,I
320 PRINT NN$;
330 LOCATE HX+1,I+1
340 PRINT RIGHT$(A$(1),2);
350 FOR J=1 TO 100
360 NEXT J
370 NEXT I
380 IF S>R THEN R=S
390 LOCATE 3,0
400 COLOR 2,0
410 PRINT "SCORE :";S,
420 PRINT "RECORD :";R;
430 FOR I=1 TO 100
440 NEXT I
450 R$=INKEY$
460 IF R$<>"" THEN 450
470 LOCATE 7,13
480 PRINT "UNE AUTRE ?";
490 R$=INKEY$
500 IF R$="" THEN 490
510 IF R$<>"N" THEN S=0:GOTO 60
520 COLOR 4,6
530 CLS
540 END
550 IF BY>23 THEN BY=0
560 LOCATE BX,B1
570 PRINT N$;
580 B1=BY
590 IF BY<>0 THEN LOCATE BX,BY:PRINT B$:
BY=BY+1
600 RETURN
```

```
610 FOR I=1 TO 10
620 NEXT I
630 RETURN
640 DEFINT A-V
650 DIM A$(1)
660 DEFGR$(0)=193,224,225,255,255,255,64
,0
670 DEFGR$(1)=248,144,249,253,255,253,24
1,96
680 DEFGR$(2)=248,144,248,252,254,252,24
0,96
690 DEFGR$(3)=40,16,56,124,124,124,56,16
700 A$(0)=CHR$(32)+GR$(0)+GR$(1)
710 A$(1)=CHR$(32)+GR$(0)+GR$(2)
720 B$=GR$(3)
730 CN=32
740 N$=CHR$(32)
750 NN$=N$+N$
760 RETURN
770 HX=0
780 HY=0
790 BX=0
800 BY=0
810 B1=BY
820 Z=0.5
830 COLOR 3,0
840 CLS
850 PRINT CHR$(14);
860 FOR I=10 TO 30
870 C=INT(RND*20)+107
880 FOR J=23 TO INT(RND*5)+14 STEP-1
890 LOCATE I,J,0
900 PRINT CHR$(C);
910 NEXT J
920 NEXT I
930 PRINT CHR$(15);
940 RETURN
```


D.C.A.

Les rôles sont maintenant inversés. Vous manœuvrez la D.C.A. et devez essayer d'abattre les avions qui passent au-dessus de vous. Pour tirer, utilisez n'importe quelle touche. Vous disposez au départ de quinze missiles. Si vous abattez dix avions, vous obtenez un bonus de dix points et dix missiles supplémentaires.

```
10 REM *****
20 REM * D.C.A. *
30 REM *****
40 CLEAR ,,9
50 GOSUB 650
60 A$=RIGHT$(A$,1)+LEFT$(A$,39)
70 B$=RIGHT$(B$,39)+LEFT$(B$,1)
80 LOCATE 0,0
90 COLOR 3,4
100 PRINT A$;
110 LOCATE 0,2
120 PRINT B$;
130 R$=INKEY$
140 IF R$<>" " AND MY=24 THEN MY=23:M$=GR
   $(4):NM=NM-1
150 COLOR 1,4
160 IF MY<>24 THEN MY=MY-2:LOCATE MX,MY:
   PRINT M$;:LOCATE MX,MY+2:PRINT 0$;
170 IF MY<>3 THEN 270
180 IF SCREEN(MX,2)=32 THEN 270
190 LOCATE MX,2
200 PRINT E$;
```

```
210 LOCATE MX,MY
220 PRINT O$;
230 BEEP
240 S=S+1
250 MID$(B$,18)=O1$
260 GOTO 360
270 IF MY<>1 THEN 390
280 IF SCREEN(MX,0)=32 THEN 390
290 LOCATE MX,0
300 PRINT E$;
310 S=S+1
320 LOCATE MX,MY
330 PRINT O$;
340 BEEP
350 MID$(A$,18)=O1$
360 IF S>1 AND S@10=S/10 THEN GOSUB 580
370 MY=24
380 M$=GR$(6)
390 LOCATE 0,24
400 COLOR 2,4
410 PRINT "S :";S;"M :";NM;
420 IF NM<1 AND MY=24 THEN 450
430 IF MY<=1 THEN :LOCATE MX,1:PRINT O$;
:MY=24:M$=GR$(6)
440 GOTO 60
450 IF S>R THEN R=S
460 LOCATE 2,10
470 PRINT "SCORE :";S,"RECORD :";R;
480 LOCATE 10,14
490 PRINT "UNE AUTRE ?"
500 R$=INKEY$
510 IF R$<>"" THEN 500
520 R$=INKEY$
530 IF R$="" THEN 520
540 IF R$<>"N" THEN 50
550 SCREEN 4,6
560 CLS
570 END
580 A$=A1$
590 B$=B1$
600 NM=NM+10
610 FOR I=1 TO 300
620 NEXT I
```

```
630 S=S+10
640 RETURN
650 DEFINT A-Z
660 SCREEN 3,4,0
670 CLS
680 DEFGR$(0)=193,224,224,255,255,255,64
,0
690 DEFGR$(1)=248,144,249,253,255,253,24
1,96
700 DEFGR$(2)=31,9,159,191,255,191,143,6
710 DEFGR$(3)=131,7,135,255,255,255,2,0
720 DEFGR$(4)=16,56,56,56,56,56,16,56
730 DEFGR$(5)=0,0,0,0,0,7,7,7
740 DEFGR$(6)=56,56,56,254,254,255,255,2
55
750 DEFGR$(7)=0,0,0,0,0,192,192,192
760 DEFGR$(8)=146,84,56,254,56,84,146,0
770 S=0
780 A$="":B$=""
790 FOR I=1 TO 5
800 A$=A$+GR$(0)+GR$(1)+"      "
810 B$=B$+GR$(2)+GR$(3)+"      "
820 NEXT I
830 A1$=A$:B1$=B$:E$=GR$(8)
840 J$=GR$(5)+GR$(6)+GR$(7)
850 MX=19:MY=24:M$=GR$(6):NM=15
860 O$=CHR$(32):O1$=O$+O$+O$+O$
870 COLOR 2,4:LOCATE MX-1,MY,0
880 PRINT J$;
890 COLOR 3,4:LOCATE 0,0
900 PRINT A$;
910 LOCATE 0,2
920 PRINT B$;
930 RETURN
```



TRACE

Deux joueurs s'affrontent pour se partager l'espace vital. Chacun doit s'efforcer, tout en se déplaçant, de ne jamais recouper sa trace ou celle de son adversaire, et de ne pas sortir du rectangle dessiné sur l'écran. Utilisez les joysticks ou les touches suivantes :


Joueur de droite : <P> , <L> , <M> et <@>

Joueur de gauche : <Z> , <Q> , <S> et <W>

```
10 REM *****
20 REM * TRACE *
30 REM *****
40 CLEAR ,,1
50 GOSUB 750
60 ONJK GOTO 130
70 D$=INKEY$
80 HB=(D$="L")-(D$="M")
90 VB=(D$="P")-(D$="@")
100 HA=(D$="Q")-(D$="S")
110 VA=(D$="Z")-(D$="W")
120 GOTO 170
130 HA=(STICK(0)=7)-(STICK(0)=3)
140 VA=(STICK(0)=1)-(STICK(0)=5)
150 HB=(STICK(1)=7)-(STICK(1)=3)
160 VB=(STICK(1)=1)-(STICK(1)=5)
170 IF HA<>0 THEN H1=HA:V1=0
180 IF VA<>0 THEN V1=VA:H1=0
190 IF HB<>0 THEN H2=HB:V2=0
200 IF VB<>0 THEN V2=VB:H2=0
```

```
210 X1=X1+H1
220 Y1=Y1+V1
230 IF SCREEN(X1,Y1)=127 THEN 350
240 LOCATE X1,Y1
250 COLOR 3
260 PRINT N$;
270 X2=X2+H2
280 Y2=Y2+V2
290 IF SCREEN(X2,Y2)=127 THEN 410
300 LOCATE X2,Y2
310 COLOR 5
320 PRINT N$;
330 BEEP
340 GOTO 60
350 F2=F2+1
360 GOSUB 690
370 IF F2=10 THEN 470
380 IF INKEY$<>" " THEN 380
390 GOSUB 870
400 GOTO 60
410 F1=F1+1
420 GOSUB 690
430 IF F1=10 THEN 540
440 IF INKEY$<>" " THEN 440
450 GOSUB 870
460 GOTO 60
470 CLS
480 COLOR 1,7
490 LOCATE 9,5
500 PRINT "LE JOUEUR DROIT GAGNE";
510 LOCATE 15,10
520 PRINT F2;"A";F1;
530 GOTO 600
540 CLS
550 COLOR 1,7
560 LOCATE 9,5
570 PRINT "LE JOUEUR GAUCHE GAGNE";
580 LOCATE 15,10
590 PRINT F1;"A";F2;
600 LOCATE 14,15
610 PRINT "UNE AUTRE ?";
620 IF INKEY$<>" " THEN 620
630 D$=INKEY$
```

```
640 IF D$="" THEN 630
650 IF D$<>"N" THEN RUN
660 CLS
670 SCREEN 4,6,6
680 END
690 FOR I=1 TO 5
700 BEEP
710 FOR J=1 TO 100
720 NEXT J
730 NEXT I
740 RETURN
750 CLS
760 SCREEN 2,0,0
770 DEFINT A-Z
780 ATTRB 1,1
790 LOCATE 6,10,0
800 PRINT "JOYSTICKS ?";
810 R$=INKEY$
820 IF R$="" THEN 810
830 JK=-(R$="O")
840 ATTRB 0,0
850 DEFGR$(0)=255,255,255,255,255,255,25
5,255
860 N$=GR$(0)
870 CLS
880 COLOR 2
890 FOR X1=0 TO 39
900 LOCATE X1,0:PRINT N$;
910 LOCATE X1,23:PRINT N$;
920 NEXT X1
930 FOR Y1=1 TO 22
940 LOCATE 0,Y1:PRINT N$;
950 LOCATE 39,Y1:PRINT N$;
960 NEXT Y1
970 LOCATE 2,24:COLOR 3
980 PRINT "JOUEUR GAUCHE :";F1;
990 LOCATE 22,24:COLOR 5
1000 PRINT "JOUEUR DROIT :";F2;
1010 X1=8:Y1=11:X2=32:Y2=11:H1=1:V1=0
1020 H2=-1:V2=0:S1=0:S2=0:RETURN
```


SLALOM

Partez aux sports d'hiver sans risque de vous casser une jambe ! Lancez-vous du haut de la piste et essayez de passer le plus grand nombre possible de portes sans heurter les piquets. Frappez n'importe quelle touche pour changer de direction.

```
10 REM *****
20 REM * SLALOM *
30 REM *****
40 CLEAR ,,4
50 GOSUB 460
60 FOR K=1 TO 300
70 X=SX*8
80 Y=SY*8
90 IF K>=16 AND (K-5)/10=(K-5)@10 THEN G
OSUB 360
100 IF K<284 AND K/10=K@10 THEN GOSUB 38
0
110 IF INKEY$<>"" THEN DX=-DX
120 S1=SX
130 SX=SX+DX
140 IF SX<2 THEN SX=2:DX=-DX
150 IF SX>37 THEN SX=37:DX=-DX
160 LOCATE S1,SY
170 PRINT B$;
180 LOCATE 0,24
190 PRINT
```

```
200 LOCATE SX,SY
210 PRINT S$(DX/2+0.5);
220 S1=SX
230 NEXT K
240 LOCATE 5,5
250 PRINT "PORTE(S) RATEE(S) :";T
260 LOCATE 5,8
270 PRINT "UNE AUTRE DESCENTE ?"
280 D$=INKEY$
290 IF D$<>" " THEN 280
300 D$=INKEY$
310 IF D$=" " THEN 300
320 IF D$<>"N" THEN RUN
330 CLS
340 SCREEN 4,6,6
350 END
360 IF POINT(X-4,Y)<>1 OR POINT(X+20,Y)<
>1 THEN IF POINT(X-12,Y)<>1 OR POINT(X+1
2,Y)<>1 THEN T=T+1:BEEP
370 RETURN
380 P=P+(INT(RND*3)-1)*8
390 IF P<2 THEN P=10
400 IF P>34 THEN P=26
410 LOCATE P,23
420 COLOR 1,7
430 PRINT P$;
440 COLOR 4,7
450 RETURN
460 DIM S$(1)
470 DEFINT A-Z
480 DEFGR$(0)=16,32,241,18,28,8,16,32
490 DEFGR$(1)=8,4,143,72,56,16,8,4
500 DEFGR$(2)=8,12,14,12,8,8,8,8
510 DEFGR$(3)=8,24,56,24,8,8,8,8
520 S$(0)=GR$(0)
530 S$(1)=GR$(1)
540 B$=CHR$(32)
550 P$=GR$(3)+B$+B$+GR$(2)
560 SCREEN 4,7,7
570 DX=-1
580 SY=8
590 SX=19
600 S1=SX
```

```
610 P=18
620 CLS
630 LOCATE 5,10,0
640 PRINT "TAPEZ UNE TOUCHE POUR COMMENC
ER"
650 N=RND
660 IF INKEY$="" THEN 650
670 CLS
680 RETURN
```




RAMASSE-MIETTES

Voici une curieuse façon d'utiliser un ordinateur : vous devez vous efforcer de ramasser le plus vite possible les miettes qui jonchent la nappe. Vous disposez de 30 secondes pour un nettoyage complet. Les miettes sont représentées par des points noirs. Le joystick ou les touches <Z> , <Q> , <S> et <W> vous permettent de contrôler votre ramasse-miettes.

```
10 REM *****
20 REM * RAMASSE-MIETTES *
30 REM *****
40 GOTO 90
50 K=STICK(0)
60 D1=(K=7)-(K=3)
70 D2=(K=1)-(K=5)
80 GOTO 170
90 GOSUB 320
100 LOCATE 0,24
110 PRINT "TEMPS :";INT(Z);"SCORE :";S;
N$;
120 IF Z<1 THEN 820
130 ON JS GOTO 50
140 D$=INKEY$
150 D1=(D$="Q")-(D$="S")
160 D2=(D$="Z")-(D$="W")
170 IF D1<>0 THEN DX=D1:DY=0
180 IF D2<>0 THEN DY=D2:DX=0
190 XP=PX+DX
200 YP=PY+DY
```

```
210 IF SCREEN(XP,YP)=127 THEN XP=PX:YP=PY
220 IF SCREEN(XP,YP)=46 THEN S=S+1:BEEP:
X=X+1
230 LOCATE PX,PY
240 PRINT N$;
250 LOCATE XP,YP
260 PRINT J$;
270 PX=XP
280 PY=YP
290 Z=Z-0.2
300 IF X=NM THEN 970
310 GOTO 100
320 CLS
330 SCREEN 0,2,6
340 DEFINT A-Y
350 ATTRB 1,1
360 LOCATE 10,10,0
370 PRINT "JOYSTICK ?";
380 ATTRB 0,0
390 IF INKEY$("<>") THEN 390
400 D$=INKEY$
410 I=RND
420 IF D$="" THEN 400
430 IF D$="0" THEN JS=1 ELSE JS=0
440 CLS
450 J$=CHR$(127)
460 M$="."
470 N$=CHR$(32)
480 S=0
490 NM=10
500 X=0
510 COLOR 4
520 FOR PX=0 TO 39
530 LOCATE PX,0
540 PRINT J$;
550 LOCATE PX,23
560 PRINT J$;
570 NEXT PX
580 FOR PY=1 TO 22
590 LOCATE 0,PY
600 PRINT J$;
610 LOCATE 39,PY
```

```
620 PRINT J$;
630 NEXT PY
640 COLOR 0
650 FOR I=1 TO NM
660 PX=INT(RND*38)+1
670 PY=INT(RND*22)+1
680 IF SCREEN(PX,PY)<>32 THEN 660
690 LOCATE PX,PY
700 PRINT M$;
710 NEXT I
720 COLOR 1
730 PX=INT(RND*38)+1
740 PY=INT(RND*22)+1
750 IF SCREEN(PX,PY)<>32 THEN 730
760 LOCATE PX,PY
770 PRINT J$;
780 Z=30
790 DX=0
800 DY=0
810 RETURN
820 FOR I=1 TO 500
830 NEXT I
840 IF INKEY$<>" " THEN 840
850 COLOR 1
860 IF S>R THEN R=S
870 LOCATE 13,10
880 PRINT "RECORD :";R;
890 LOCATE 13,16
900 PRINT "UNE AUTRE ?";
910 D$=INKEY$
920 IF D$="" THEN 910
930 IF D$<>"N" THEN 90
940 CLS
950 SCREEN 4,6,6
960 END
970 NM=NM+1
980 LOCATE XP,YP
990 PRINT N$;
1000 GOSUB 500
1010 GOTO 100
```


TANK

Essayez de suivre la route avec votre tank tout en évitant les mines qui s'y trouvent placées au hasard par l'ordinateur. Pour vous diriger, utilisez le joystick ou les touches < , > et < . > . Si le passage est bloqué, vous pouvez faire sauter une ou plusieurs mines en tirant un de vos dix obus (frappez la barre d'espace ou appuyez sur le bouton du joystick).

```
10 REM *****
20 REM * TANK *
30 REM *****
40 CLEAR ,,2
50 GOSUB 740
60 GOTO 120
70 TX=TX+(STICK(0)=7)-(STICK(0)=3)
80 RETURN
90 X$=INKEY$
100 TX=TX+(X$=G$)-(X$=D$)
110 RETURN
120 ON JS GOSUB 70,90
130 LOCATE X,Y
140 PRINT N$;
150 LOCATE 0,24
160 COLOR 0
170 PRINT R$
180 IF POINT(TX*8+4,TY*8+4)<>-4 THEN 430
190 LOCATE TX,TY
200 COLOR 1
```

```
210 PRINT T$;
220 IF X$=N$ OR (STRIG(0)=-1 AND JS=1) T
HEN GOSUB 360
230 X=TX
240 Y=TY
250 COLOR 0
260 PO=INT(RND*7)+17
270 LOCATE PO,23
280 PRINT M$;
290 PO=INT(RND*7)+17
300 LOCATE PO,23
310 PRINT M$;
320 S=S+1
330 FOR I=1 TO DL-S
340 NEXT I
350 GOTO 120
360 IF OB=0 THEN 420
370 OB=OB-1
380 FOR I=1 TO 5
390 LOCATE TX,TY+I
400 PRINT N$;
410 NEXT I
420 RETURN
430 COLOR 1
440 FOR I=1 TO 10
450 LOCATE TX,TY
460 PRINT M$;
470 FOR J=1 TO 100
480 NEXT J
490 LOCATE TX,TY
500 PRINT T$;
510 FOR J=1 TO 100
520 NEXT J
530 NEXT I
540 IF S>R THEN R=S
550 CLS
560 COLOR 0
570 ATTRB 1,1
580 LOCATE 10,10
590 PRINT "SCORE :";S;
600 LOCATE 9,14
610 PRINT "RECORD :";R;
620 LOCATE 9,18
```

```
630 PRINT "UNE AUTRE ?";
640 ATTRB 0,0
650 IF INKEY$<>"" THEN 650
660 X$=INKEY$
670 IF X$="" THEN 660
680 IF X$<>"N" THEN 720
690 CLS
700 SCREEN 4,6,6
710 END
720 GOSUB 1010
730 GOTO 120
740 CLS
750 SCREEN 2,3,3
760 DEFINT A-Z
770 DEFGR$(0)=198,254,254,254,254,214,21
4,16
780 DEFGR$(1)=0,84,56,124,56,84,0,0
790 T$=GR$(0):M$=GR$(1):N$=CHR$(32)
800 N$=CHR$(32)
810 R$=""
820 FOR I=1 TO 15
830 R$=R$+N$
840 NEXT I
850 R$=R$+M$+M$
860 FOR I=1 TO 7
870 R$=R$+N$
880 NEXT I
890 R$=R$+M$+M$
900 D$="." : G$="," : LOCATE 10,10,0
910 ATTRB 1,1:PRINT "JOYSTICK ?";
920 X$=INKEY$
930 IF X$="" THEN 920
940 IF X$="O" THEN JS=1 ELSE JS=2
950 CLS
960 LOCATE 5,10
970 PRINT "TAPEZ UNE TOUCHE";
980 PO=RND
990 IF INKEY$="" THEN 980
1000 ATTRB 0,0
1010 CLS:TX=19:TY=10:X=TX:Y=TY:S=0
1020 DL=200:OB=10:COLOR 0
1030 FOR I=1 TO 24:PRINT R$:NEXT I
1040 RETURN
```



ALPHABET



Voici un jeu d'adresse assez difficile. Vous tenterez de marquer le plus grand nombre de points possible en effaçant les lettres affichées par l'ordinateur. Utilisez le joystick ou les touches < Z > (haut), < W > (bas), < Q > (gauche) et < S > (droite).

Attention : vous devez effacer les lettres dans l'ordre alphabétique en évitant les obstacles placés au hasard sur l'écran, et ceci en un temps limité. (Le temps qui vous reste est affiché au bas de l'écran.) Lorsque toutes les lettres sont effacées, le jeu reprend avec une lettre supplémentaire.

```
10 REM *****
20 REM * ALPHABET *
30 REM *****
40 GOSUB 920
50 S=0
60 X=0
70 N$=CHR$(32)
80 P$=CHR$(127)
90 GOSUB 640
100 FOR I=1 TO X
110 ON JK GOTO 180
120 D$=INKEY$
130 DH=(D$="Q")-(D$="S")
140 DV=(D$="Z")-(D$="W")
150 IF DH<>0 THEN DX=DH: DY=0
160 IF DV<>0 THEN DY=DV: DX=0
170 GOTO 210
180 K=STICK(0)
```

```
190 DX=(K=7)-(K=3)
200 DY=(K=1)-(K=5)
210 Z=Z-0.1
220 LOCATE 0,24
230 COLOR 0
240 PRINT "TEMPS :";INT(Z);
250 IF Z<1 THEN 460
260 PX=PX+DX
270 PY=PY+DY
280 C=SCREEN(PX,PY)
290 IF C=I+64 THEN S=S+1:BEEP:GOTO 380
300 IF C<>32 THEN PX=XP:PY=YP
310 COLOR 1
320 LOCATE XP,YP
330 PRINT N$;
340 LOCATE PX,PY
350 PRINT P$;
360 XP=PX:YP=PY
370 GOTO 110
380 LOCATE XP,YP
390 PRINT N$;
400 LOCATE PX,PY
410 PRINT P$;
420 XP=PX:YP=PY
430 NEXT I
440 GOSUB 640
450 GOTO 100
460 D$=INKEY$
470 IF R<S THEN R=S
480 COLOR 0
490 LOCATE 10,10
500 PRINT "TEMPS ECOULE";
510 LOCATE 10,13
520 IF INKEY$<>" " THEN 520
530 COLOR 0
540 PRINT "SCORE :";S;
550 LOCATE 10,16
560 PRINT "RECORD :";R;
570 LOCATE 10,19
580 PRINT "UNE AUTRE ?";
590 D$=INKEY$
600 IF D$="" THEN 590
610 IF D$<>"N" THEN 50
```

```
620 CLS:SCREEN 4,6,6
630 END
640 CLS
650 X=X+1
660 COLOR 0
670 FOR PX=0 TO 39
680 LOCATE PX,0:PRINT P$;
690 LOCATE PX,23:PRINT P$;
700 NEXT PX
710 FOR PY=1 TO 22
720 LOCATE 0,PY:PRINT P$;
730 LOCATE 39,PY:PRINT P$;
740 NEXT PY
750 FOR Z=1 TO 120
760 GOSUB 890
770 LOCATE PX,PY:PRINT P$;
780 NEXT Z
790 FOR Z=1 TO X
800 GOSUB 890
810 COLOR 3,5:LOCATE PX,PY
820 PRINT CHR$(Z+64);
830 NEXT Z
840 COLOR 1,2
850 GOSUB 890
860 LOCATE PX,PY:PRINT P$;
870 XP=PX:YP=PY:Z=30:DX=0:DY=0
880 RETURN
890 PX=INT(RND*38)+1:PY=INT(RND*23)+1
900 IF SCREEN(PX,PY)<>32 THEN 890
910 RETURN
920 SCREEN 4,2,4:DEFINT A-Y:ATTRB 1,1
930 CLS:LOCATE 10,10,0
940 PRINT "JOYSTICK ?";:ATTRB 0,0
950 D$=INKEY$:C=RND
960 IF D$="" THEN 950
970 IF D$="0" THEN JK=1
980 RETURN
```



ATTERRISSAGE

Après un long voyage en apesanteur, poser une navette spatiale en douceur n'est pas chose aisée ; mais grâce à votre ordinateur, vous allez être en mesure de vous entraîner sans danger. Vous devez poser votre navette sur l'aire prévue à cet effet. Vous pouvez vous diriger vers la droite et vers la gauche à l'aide des touches <, > et <.> ou freiner votre descente avec la barre d'espacement. (Vous pouvez également utiliser le joystick.) La quantité de carburant disponible est indiquée par la longueur du trait horizontal, en haut de l'écran.

```
10 REM *****
20 REM * ATTERRISSAGE *
30 REM *****
40 DEFINT A-Z
50 DIM N(15,20)
60 DIM R(15,20)
70 DIM C(15,10)
80 FU=100
90 GOTO 620
100 GOSUB 750
110 DV=DV+1
120 IF FU<1 THEN D$="":GOTO 170
130 D$=INKEY$
140 Q=STICK(0)
150 DV=DV+3*(Q=5)
160 DH=DH+(Q=3)-(Q=7)
170 DV=DV+3*(D$=" ")
180 DH=DH+(D$=".")-(D$=","")
```

```
190 IF D$<>" " OR Q<>0 THEN FU=FU-1:FF=1
200 DRAW "COL"+STR$(FF)
210 FF=0
220 IF MH<0 OR NH>319 THEN 250
230 IF MV<0 THEN 250
240 PUT (MH,MV)-(NH,NV),R
250 NH=NH+DH
260 NV=NV+DV
270 IF NV>187 THEN 340
280 MV=NV-20
290 MH=NH-15
300 IF MH<0 OR NH>319 THEN 110
310 IF MV<0 THEN 110
320 PUT (MH,MV)-(NH,NV),N
330 GOTO 110
340 IF DV>5 THEN 420
350 IF ABS(NH-A)>4 THEN 420
360 IF ABS(DH)>1 THEN 420
370 PUT (MH,167)-(NH,187),N
380 FOR I=0 TO 1000
390 NEXT I
400 S=S+10
410 GOTO 100
420 IF MH<0 OR NH>304 THEN 460
430 IF MV<0 OR NV>191 THEN 450
440 PUT (MH,MV)-(NH,NV),R
450 PUT (MH,181)-(NH,191),C
460 FOR I=1 TO 10
470 NEXT I
480 CLS
490 SCREEN 4,6,6
500 LOCATE 5,10
510 PRINT "VOTRE NAVETTE S'EST ECRASEE"
520 LOCATE 10,13
530 PRINT "SCORE :";S;
540 LOCATE 10,16
550 PRINT "UNE AUTRE ?";
560 IF INKEY$<>" " THEN 560
570 D$=INKEY$
580 IF D$="" THEN 570
590 IF D$<>"N" THEN RUN
600 CLS
610 END
```

```
620 CLS
630 SCREEN 3,0,2
640 LOCATE 0,0,0
650 FOR I=1 TO 15
660 FOR J=0 TO 10
670 C(I,J)=RND*1000
680 NEXT J
690 NEXT I
700 DRAW "C3BM50,50R2L1U3E3H4U6E4R7F4D6G
4F3D3L1R2L1U3H3L3D2R1D1U1L3D1U1R1U2L3"
710 PAINT (55,40),3
720 GET (50,30)-(65,50),N
730 GET (0,0)-(15,20),R
740 GOTO 100
750 CLS
760 A=INT(RND*260)+20
770 DRAW "C1BM"+STR$(A)+",188R4L23D1R23D
1L23"
780 NH=160
790 NV=20
800 DH=0
810 DV=0
820 MV=0
830 MH=NH-15
840 FU=FU+10
850 DRAW "C1BM0,3R"+STR$(FU)
860 RETURN
```



CHASSE AU CANARD

Votre microordinateur vous invite à une partie de chasse au canard. Les canards volent de droite à gauche en haut de l'écran. Vous vous déplacez en utilisant le joystick ou les touches < Q > et < S > . La touche < W > vous permet de vous arrêter. Vous pouvez tirer autant de cartouches que vous voulez, mais il ne passera que 20 canards, et votre but est d'en abattre le plus possible. Pour tirer, frappez la barre d'espacement ou pressez le bouton du joystick. Si vous atteignez un canard, vous marquerez un point et vous le verrez tomber en battant des ailes.



```
10 REM *****
20 REM * CHASSE AU CANARD *
30 REM *****
40 CLEAR ,,6
50 GOSUB 410
60 Z=-Z
70 H=Z+0.5
80 CX=CX-1
90 IF CX<0 THEN GOSUB 750
100 IF NC=0 THEN 900
110 LOCATE CX,CY
120 COLOR 2
```


```
130 PRINT CC$(H);
140 JX=JX+D1
150 IF JX>38 THEN JX=38
160 IF JX<0 THEN JX=0
170 LOCATE X,Y
180 PRINT N$;
190 LOCATE JX,JY
200 COLOR 1
210 PRINT J$;
220 X=JX
230 IF TY<>JY THEN TY=TY-2:GOTO 250
240 TX=JX
250 IF TY<2 THEN 800
260 LOCATE TX,TY
270 COLOR 1
280 PRINT J$;
290 IF TY<>JY THEN LOCATE TX,TY+2:PRINT
N$;
310 X$=INKEY$
320 D=2*((X$="Q")-(X$="S"))
330 IF D<>0 THEN D1=D
340 IF X$="W" THEN D1=0
350 JX=JX+2*((STICK(0)=7)-(STICK(0)=3))
360 IF X$<>N$ AND (JS=0 OR STRIG(0)=0) T
HEN 60
370 IF TY<>JY THEN 60
380 TY=JY-2
390 TX=JX
400 GOTO 60
410 CLS
420 DEFINT A-Y
430 DEFGR$(0)=0,0,0,15,31,0,0,0
440 DEFGR$(1)=63,124,254,255,255,0,0,0
450 DEFGR$(2)=0,0,254,255,255,124,124,15
460 DEFGR$(3)=0,0,0,128,240,0,0,0
470 DEFGR$(4)=8,28,42,8,8,8,8,20
480 DEFGR$(5)=146,84,56,254,56,84,146,0
490 SCREEN 2,0,0
500 LOCATE 10,10,0
510 ATTRB 1,1
520 PRINT "JOYSTICK ?";
530 ATTRB 0,0
540 X$=INKEY$
```

```
550 IF X$="" THEN 540
560 IF X$="0" THEN JS=1 ELSE JS=0
570 CLS
580 DIM CC$(1)
590 J$=GR$(4)
600 N$=CHR$(32)
610 E$=GR$(5)
620 CC$(0)=GR$(0)+GR$(1)+GR$(3)+N$
630 CC$(1)=GR$(0)+GR$(2)+GR$(3)+N$
640 JX=19
650 JY=24
660 NC=20
670 TX=JX
680 TY=JY
690 Z=0.5
700 CX=36
710 CY=0
720 X=JX
730 Y=JY
740 RETURN
750 LOCATE 0,0
760 PRINT N$+N$+N$;
770 CX=36
780 NC=NC-1
790 RETURN
800 IF TX=CX OR TX=CX+1 THEN 1040
810 LOCATE TX,TY+2
820 PRINT N$;
830 LOCATE TX,TY
840 PRINT J$;
850 LOCATE TX,TY
860 PRINT N$;
870 TX=JX
880 TY=JY
890 IF NC<>0 THEN 260
900 CLS
910 ATTRB 1,1
920 LOCATE 10,10
930 PRINT "SCORE :";S;
940 LOCATE 9,20
950 PRINT "UNE AUTRE ?";
960 ATTRB 0,0
970 IF INKEY$<>"" THEN 970
```



```
980 X$=INKEY$
990 IF X$="" THEN 980
1000 IF X$<>"N" THEN RUN
1010 CLS
1020 SCREEN 4,6,6
1030 END
1040 LOCATE TX,TY
1050 PRINT E$;
1060 LOCATE TX,TY+2
1070 PRINT N$;
1080 BEEP
1090 FOR I=1 TO 50
1100 NEXT I
1110 S=S+1
1120 DX=CX
1130 CX=36
1140 COLOR 2
1150 FOR I=1 TO 24
1160 Z=-Z
1170 H=Z+0.5
1180 LOCATE DX,I-1
1190 PRINT N$+N$+N$;
1200 LOCATE DX,I
1210 PRINT CC$(H);
1220 BEEP
1230 FOR J=1 TO 10
1240 NEXT J
1250 NEXT I
1260 NC=NC-1
1270 FOR I=1 TO 500
1280 NEXT I
1290 CLS
1300 IF NC=0 THEN 900
1310 TY=JY
1320 GOTO 60
```






SERPENT

Dans ce jeu, vous êtes un serpent qui se déplace en ondulant sur l'écran. Le changement de direction s'effectue en tapant n'importe quelle touche. Pour pouvoir vous déplacer, vous devez vous nourrir. Heureusement, vous êtes entouré par un grand nombre de champignons. Mais attention ! Si les bleus sont excellents, vous devez absolument éviter les rouges qui, eux, sont vénéneux. Chaque champignon bleu vous apporte suffisamment de calories pour avancer de dix lignes. Essayez de ne pas mourir de faim sans pour autant finir empoisonné !

```
10 REM *****
20 REM * SERPENT *
30 REM *****
40 CLEAR ,,2
50 GOSUB 540
60 D$=INKEY$
70 IF D$<>" " THEN D=-D
80 X=X+D
90 IF X<0 THEN X=1
100 IF X>39 THEN X=38
110 IF POINT(X*8+4,Y*8+4)=1 THEN 260
120 IF POINT(X*8+4,Y*8+4)=4 THEN S=S+10:
H=H+10:BEEP
130 LOCATE X,Y
140 COLOR 0
```

```
150 PRINT S$;
160 LOCATE 0,24,0
170 PRINT
180 LOCATE INT(RND*38)+1,24
190 COLOR 4
200 PRINT C$;
210 IF RND>0.5 THEN LOCATE INT(RND*38)+1
,24:COLOR 1:PRINT C$;
220 S=S-1
230 IF S=0 THEN 260
240 H=H+1
250 GOTO 60
260 BEEP
270 LOCATE 0,24
280 PRINT
290 LOCATE X,Y-1
300 COLOR 5
310 PRINT S$;
320 FOR I=1 TO 5
330 BEEP
340 FOR J=1 TO 50
350 NEXT J
360 NEXT I
370 IF INKEY$<>" " THEN 370
380 IF H>R THEN R=H
390 LOCATE 4,20
400 COLOR 0
410 PRINT "SCORE :";H;
420 LOCATE 23,20
430 PRINT "RECORD :";R;
440 LOCATE 15,23
450 PRINT "UNE AUTRE ?"
460 D$=INKEY$
470 IF D$="" THEN 460
480 IF D$<>"N" THEN 520
490 SCREEN 4,6,6
500 CLS
510 END
520 GOSUB 610
530 GOTO 60
540 CLS
550 SCREEN 2,2,2
560 DEFINT A-Z
```

```
570 DEFGR$(0)=0,0,102,255,219,126,60,24
580 DEFGR$(1)=60,126,126,255,255,24,24,6
0
590 S$=GR$(0)
600 C$=GR$(1)
610 CLS
620 S=100
630 H=0
640 D=1
650 X=19
660 Y=10
670 RETURN
```



SIMON

Dans cette version du jeu bien connu, vous utiliserez les touches de déplacement du curseur pour répéter la séquence produite de façon aléatoire par l'ordinateur. A chaque touche correspondent une couleur et un son :

(du plus grave au plus aigu)

< ↑ > : rouge

< ↓ > : bleu

< → > : jaune

< ← > : violet


La limite de la longueur de la séquence à répéter est 100. Lorsque vous faites une erreur, le jeu s'arrête et votre score, correspondant à la longueur de la plus longue séquence que vous avez réussi à reproduire, est affiché ainsi que le record du jour.

```
10 REM *****
20 REM * SIMON *
30 REM *****
40 DIM T(100)
50 X=0
60 SCREEN 2,2,2
70 ATTRB 0,0
80 CLS
90 GOSUB 730
100 X=X+1
110 T(X)=INT(RND*4)+1
120 FOR I=1 TO X
130 ON T(I) GOSUB 590,660,520,450
140 NEXT I
```



```
150 FOR I=1 TO X
160 FOR J=1 TO 1000
170 D$=INKEY$
180 IF D$<>" " THEN 210
190 NEXT J
200 GOTO 280
210 Z=ASC(D$)-7
220 IF Z<>T(I) THEN 280
230 ON Z GOSUB 590,660,520,450
240 NEXT I
250 FOR I=1 TO 500
260 NEXT I
270 GOTO 100
280 X=X-1
290 IF X>R THEN R=X
300 IF INKEY$<>" " THEN 300
310 COLOR 0
320 LOCATE 10,11
330 PRINT "SCORE :";X;
340 LOCATE 10,15
350 PRINT "RECORD :";R;
360 LOCATE 10,19
370 PRINT "UNE AUTRE ?";
380 D$=INKEY$
390 IF D$="" THEN 380
400 IF D$<>"N" THEN 50
410 CLS
420 SCREEN 4,6,6
430 ATTRB 0,0
440 END
450 LOCATE 13,11
460 COLOR 1
470 PRINT R$;
480 PLAY "D0"
490 LOCATE 13,11
500 PRINT N$;
510 RETURN
520 LOCATE 23,11
530 COLOR 4
540 PRINT R$;
550 PLAY "MI"
560 LOCATE 23,11
570 PRINT N$;
```

```
580 RETURN
590 LOCATE 13,19
600 COLOR 3
610 PRINT R$;
620 PLAY "SO"
630 LOCATE 13,19
640 PRINT N$;
650 RETURN
660 LOCATE 23,19
670 COLOR 5
680 PRINT R$;
690 PLAY "SI"
700 LOCATE 23,19
710 PRINT N$;
720 RETURN
730 COLOR 5
740 FOR I=0 TO 39
760 LOCATE I,0,0:PRINT CHR$(127);
780 LOCATE I,23:PRINT CHR$(127);
790 NEXT I
800 FOR I=1 TO 22
820 LOCATE 0,I:PRINT CHR$(127);
840 LOCATE 39,I:PRINT CHR$(127);
850 NEXT I
860 FOR I=13 TO 24
880 LOCATE I,2:PRINT CHR$(127);
900 LOCATE I,5:PRINT CHR$(127);
910 NEXT I
920 FOR I=3 TO 4
940 LOCATE 13,I:PRINT CHR$(127);
960 LOCATE 24,I:PRINT CHR$(127);
970 NEXT I
980 ATTRB 1,1:LOCATE 14,4
1000 COLOR 1:PRINT "SIMON";
1020 ATTRB 0,0:LOCATE 10,10
1040 PRINT "TAPEZ UNE TOUCHE...";
1050 I=RND
1060 IF INKEY$="" THEN 1050
1070 LOCATE 10,10
1080 PRINT " ";
1090 ATTRB 1,1:R$=CHR$(127):N$=CHR$(32)
1120 RETURN
```




GRAND-PRIX

Au volant de votre formule 1, essayez de parcourir la plus grande distance possible. Votre voiture dispose de deux vitesses que vous pouvez sélectionner à l'aide des touches <1> et <2> si vous utilisez le clavier. Si vous disposez d'un joystick, maintenez le bouton enfoncé pour rester en seconde vitesse. La direction est commandée par les touches < , > et < . > ou par le joystick. En seconde vitesse, votre voiture roule deux fois plus vite. Mais gare à l'accident !

```
10 REM *****
20 REM * GRAND-PRIX *
30 REM *****
40 CLEAR ,,2
50 GOSUB 590
60 ON JS GOTO 120
70 D$=INKEY$
80 IF D$="1" THEN T=1
90 IF D$="2" THEN T=2
100 VX=VX+(D$=",")-(D$=".")
110 GOTO 150
120 H=STICK(0)
130 VX=VX+(H=7)-(H=3)
140 IF STRIG(0) THEN T=2 ELSE T=1
150 IF SCREEN(VX,VY+1)<>127 THEN 330
160 RX=RX+(RND<0.5)-(RND<0.5)
170 IF RX<RN THEN RX=RN
180 IF RX>RM THEN RX=RM
190 LOCATE XV,VY
200 COLOR 0,2
```

```
210 PRINT N$;
220 LOCATE RX,24
230 PRINT R$
240 LOCATE VX,VY
250 COLOR 1,0
260 PRINT V$;
270 K=K+T
280 DL=(2-T)*50
290 FOR I=1 TO DL
300 NEXT I
310 XV=VX
320 GOTO 60
330 LOCATE XV,VY
340 COLOR 0
350 PRINT N$
360 FOR I=1 TO 5
370 LOCATE VX,VY+1
380 COLOR 1,2
390 PRINT A$;
400 FOR J=1 TO 100
410 NEXT J
420 BEEP
430 LOCATE VX,VY+1
440 PRINT V$;
450 FOR J=1 TO 100
460 NEXT J
470 NEXT I
480 LOCATE 10,10
490 PRINT "KMS PARCOURUS :";K;
500 IF INKEY$<>"" THEN 500
510 LOCATE 14,16
520 PRINT "UNE AUTRE ?";
530 D$=INKEY$
540 IF D$="" THEN 530
550 IF D$<>"N" THEN RUN
560 CLS
570 SCREEN 4,6,6
580 END
590 CLS
600 SCREEN 0,2,2
610 DEFINT A-Y
620 ATTRB 1,1
630 LOCATE 10,10,0
```

```
640 PRINT "JOYSTICK ?"
650 ATTRB 0,0
660 D$=INKEY$
670 IF D$="" THEN 660
680 IF D$="0" THEN JS=1
690 CLS
700 N$=CHR$(127)
710 DEFGR$(0)=219,255,219,24,24,90,126,9
0
720 DEFGR$(1)=146,84,56,254,56,84,146,0
730 A$=GR$(1)
740 R$=N$+N$+N$
750 RX=18
760 RY=24
770 T=1
780 V$=GR$(0)
790 VX=19
800 VY=10
810 XV=VX
820 LOCATE 0,0
830 FOR I=0 TO 24
840 LOCATE RX,I
850 PRINT R$
860 NEXT I
870 LOCATE VX,VY
880 COLOR 1,0
890 PRINT V$;
900 RM=35
910 RN=0
920 RETURN
```



CRABES

Vous devez maintenant aider une pauvre crevette à regagner la mer en évitant les crabes voraces qui patrouillent sur la plage. Chaque crevette amenée au but rapporte un point. Vous disposez de cinq vies pour tenter de marquer un score maximum. Utilisez les touches < Z > pour avancer et < W > pour reculer.


```
10 REM *****
20 REM * CRABES *
30 REM *****
40 CLEAR ,3
50 GOSUB 840
60 LOCATE 0,20
70 COLOR 0
80 PRINT "VIE(S) REST.";NF;
90 A$=RIGHT$(A$,1)+LEFT$(A$,39)
100 B$=RIGHT$(B$,39)+LEFT$(B$,1)
110 LOCATE 0,X1
120 PRINT A$;
130 LOCATE 0,X2
140 PRINT B$;
150 LOCATE 0,X3
160 PRINT A$;
170 LOCATE 0,X4
```




```
180 PRINT B$;
190 D$=INKEY$
200 PY=PY+(D$="Z")-(D$="W")
210 IF PY>10 THEN PY=10
220 IF PY=2 THEN 350
230 C=POINT (PX*8+3,PY*8+3)
240 IF C=0 THEN 550
250 LOCATE PX,YP
260 PRINT N$;
270 LOCATE PX,PY
280 COLOR 1
290 PRINT P$;
300 COLOR 0
310 YP=PY
320 T=T+1
330 IF T>500 THEN 660
340 GOTO 60
350 LOCATE PX,YP
360 PRINT N$;
370 LOCATE PX,PY
380 COLOR 1
390 PRINT P$;
400 COLOR 0
410 BEEP
420 FOR I=1 TO 200
430 NEXT I
440 LOCATE PX,PY
450 PRINT N$;
460 PY=10
470 YP=PY
480 S=S+1
490 LOCATE 0,0
500 PRINT "SCORE :";S;
510 LOCATE 19,0
520 PRINT "RECORD :";R;
530 GOSUB 1110
540 GOTO 60
550 NP=NP-1
560 LOCATE PX,YP
570 PRINT N$;
580 LOCATE PX,PY
590 PRINT CHR$(127);
600 GOSUB 1160
```

```
610 IF NP=0 THEN 660
620 PY=10
630 YP=PY
640 GOSUB 1110
650 GOTO 60
660 CLS
670 IF S>R THEN R=S
680 IF T<500 THEN 710
690 LOCATE 10,8
700 PRINT "** TEMPS ECOULE **";
710 LOCATE 10,12
720 PRINT "SCORE :";S;
730 LOCATE 10,16
740 PRINT "RECORD :";R;
750 LOCATE 10,20
760 PRINT "UNE AUTRE ?";
770 IF INKEY$<>"" THEN 770
780 D$=INKEY$
790 IF D$="" THEN 780
800 IF D$<>"N" THEN 50
810 CLS
820 SCREEN 4,6,6
830 END
840 CLS
850 LOCATE 0,0,0
860 SCREEN 0,3,3
870 DEFINT A-Z
880 DEFGR$(0)=1,2,4,12,28,56,112,127
890 DEFGR$(1)=7,15,31,31,18,16,12,0
900 DEFGR$(2)=224,240,248,248,72,8,48,0
910 F$=GR$(0)
920 N$=CHR$(32)
930 A$=""
940 B$=""
950 S=0
960 NP=5
970 PX=19
980 PY=10
990 YP=PY
1000 X1=4
1010 X2=5
1020 X3=7
1030 X4=8
```

```
1040 T=0
1050 RESTORE
1060 FOR I=1 TO 40
1070 READ A
1080 A$=A$+CHR$(A)
1090 NEXT I
1100 B$=A$
1110 X=RND(38)+1
1120 A$=RIGHT$(A$,X)+LEFT$(A$,40-X)
1130 RETURN
1140 DATA 32,129,130,32,32,129,130,32,32
,32,129,130,32,32,32,32,32,129,130,32
1150 DATA 32,32,129,130,32,32,32,129,130
,32,32,129,130,32,32,129,130,32,32,32,32
1160 PLAY "T1003L8S004L90D003L16S004L8D0
L90MIT5L20D0L10MIL20S0L10FA#L20FAL10RE#L
20MIL10D003L20LAL10S004L60D0L10MI"
1170 FOR I=1 TO 200
1180 NEXT I
1190 RETURN
```





CASSE-BRIQUES

Voici un jeu que l'on a l'habitude de voir dans les cafés. Grâce à votre ordinateur, vous allez maintenant pouvoir en profiter sans dépenser un centime. Le but du jeu est simple : essayer de détruire un mur de briques à l'aide d'une balle que vous devez renvoyer avec votre raquette. Chaque brique cassée rapporte un point. Lorsque le mur est entièrement détruit, un nouveau mur apparaît. Vous disposez de dix balles pour tenter de marquer un score maximum. Utilisez le joystick ou les touches < Q > , < S > et la barre d'espace pour déplacer la raquette.


```
10 REM *****
20 REM * CASSE-BRIQUES *
30 REM *****
40 CLEAR , , 4
50 GOSUB 730
60 GOTO 200
70 D=2*((STICK(0)=7)-(STICK(0)=3))
80 IF D<>0 THEN DO=D
90 IF STICK(0)=0 THEN DO=0
100 RX=RX+DO
110 CH=SGN(DO)
120 RETURN
130 D$=INKEY$
140 D=2*((D$="Q")-(D$="S"))
150 IF D<>0 THEN DO=D
160 IF D$=N$ THEN DO=0
170 RX=RX+DO
180 CH=SGN(DO)
```


```
190 RETURN
200 LOCATE BX,BY
210 COLOR 1,7
220 PRINT N$;
230 BX=BX+DX
240 BY=BY+DY
250 IF SCREEN(BX,BY)<>32 THEN DY=-DY:S=S
+1:BEEP
260 LOCATE BX,BY
270 PRINT B$;
280 IF BY=22 AND ABS(BX-RX-3)>1 THEN 410
290 IF BY=22 AND BX>3 AND BX<36 THEN LOC
ATE BX,BY:PRINT N$;:BX=BX+CH
300 IF BY=22 THEN BEEP:DY=-DY
310 IF BY=1 THEN BEEP:DY=-DY
320 IF BX=2 OR BX=37 THEN BEEP:DX=-DX
330 ON JS GOSUB 70,130
340 IF RX<0 THEN RX=0
350 IF RX>33 THEN RX=33
360 LOCATE RX,RY
370 COLOR 0
380 PRINT R$;
390 IF S=204 THEN GOSUB 1210
400 GOTO 200
410 NB=NB+1
420 IF NB=11 THEN 530
430 LOCATE BX,BY
440 PRINT N$;
450 FOR I=1 TO 3
460 BEEP
470 FOR J=1 TO 100
480 NEXT J
490 NEXT I
500 DO=0
510 GOSUB 1150
520 GOTO 200
530 LOCATE 13,5
540 COLOR 0
550 S=S+404*K
560 PRINT "SCORE :";S;
570 IF S>R1 THEN R1=S
580 LOCATE 13,10
590 PRINT "RECORD :";R1;
```

```
600 LOCATE 13,15
610 PRINT "UNE AUTRE ?";
620 NB=0
630 K=0
640 S=0
650 D$=INKEY$
660 IF D$<>" " THEN 650
670 D$=INKEY$
680 IF D$="" THEN 670
690 IF D$<>"N" THEN 50
700 SCREEN 4,6,6
710 CLS
720 END
730 CLS
740 SCREEN 1,7,7
750 ATTRB 1,1
760 DEFINT A-Z
770 DEFGR$(0)=255,255,255,0,0,0,0,0
780 DEFGR$(1)=60,126,255,255,255,255,126
,60
790 DEFGR$(2)=8,8,8,8,8,8,8,255
800 DEFGR$(3)=0,0,0,0,0,0,0,255
810 B1$=GR$(2)
820 B2$=GR$(3)
830 N$=CHR$(32)
840 R$=N$+N$+GR$(0)+GR$(0)+GR$(0)+N$+N$
850 B$=GR$(1)
860 LOCATE 1,10,0
870 PRINT "JOYSTICK (O ou N) ?";
880 D$=INKEY$
890 IF D$="" THEN 880
900 IF D$="O" THEN JS=1 ELSE JS=2
910 CLS
920 ATTRB 0,0
930 COLOR 7,4
940 FOR BX=1 TO 38
950 LOCATE BX,0
960 PRINT N$;
970 NEXT BX
980 FOR BY=1 TO 22
990 LOCATE 1,BY
1000 PRINT N$;
1010 LOCATE 38,BY
```



```
1020 PRINT N$;
1030 NEXT BY
1040 FOR BY=2 TO 7
1050 COLOR 0,BY-1
1060 FOR BX=3 TO 35 STEP 2
1070 LOCATE BX,BY
1080 IF BY/2<>INT(BY/2) THEN 1110
1090 PRINT B1$;B2$;
1100 GOTO 1120
1110 PRINT B2$;B1$;
1120 NEXT BX
1130 NEXT BY
1140 RX=16
1150 RY=23
1160 BY=22
1170 BX=INT(RND*34)+3
1180 DY=-1
1190 DX=(INT(RND*2)-0.5)*2
1200 RETURN
1210 K=K+1
1220 S=0
1230 LOCATE BX,BY
1240 PRINT N$;
1250 LOCATE RX+2,RY
1260 PRINT N$;N$;N$;
1270 GOSUB 930
1280 RETURN
```





POURSUITE

Le voleur s'est échappé, emportant le magot. (Il est représenté par un masque noir.) Il se cache dans la ville et vous avez trente minutes pour le débusquer et l'arrêter. Attention, pas de précipitation ! En effet, si vous vous jetez sur lui sans réfléchir, il y a toutes les chances pour qu'il vous file entre les doigts. La meilleure façon de vous y prendre est de l'aborder de côté. (Efficace à tous les coups à condition de ne pas le rater !) Si vous ne vous sentez pas assez sûr de vous, attaquez-le de face, ce qui est plus facile mais moins efficace car moins discret. Encore un conseil : n'essayez pas de le poursuivre ; cela ne vous mènerait à rien car il est aussi rapide que vous. Observez plutôt ses mouvements comme un vrai détective. Quand vous le verrez tourner en rond, approchez-vous sans faire de bruit et surgissez au bon moment. Mais rappelez-vous, le temps presse !

Pour vous déplacer, utilisez le joystick ou les touches suivantes :

< Z > : haut
< Q > : gauche
< S > : droite
< W > : bas


```
10 REM *****
20 REM * POURSUITE *
30 REM *****
40 CLEAR , , 2
50 GOSUB 1320
60 S=0
70 N$=CHR$(32)
```

```
80 V$=GR$(1)
90 P$=GR$(0)
100 GOSUB 820
110 ON JK GOTO 180
120 D$=INKEY$
130 DH=(D$="Q")-(D$="S")
140 DV=(D$="Z")-(D$="W")
150 IF DH<>0 THEN DX=DH: DY=0
160 IF DV<>0 THEN DY=DV: DX=0
170 GOTO 210
180 ST=STICK(0)
190 DX=(ST=7)-(ST=3)
200 DY=(ST=1)-(ST=5)
210 Z=Z-0.2
220 LOCATE 0,24
230 COLOR 0
240 PRINT "TEMPS : ";INT(Z+1);
250 IF Z<0 THEN 500
260 PX=PX+DX
270 PY=PY+DY
280 C=POINT(PX*8+4,PY*8+4)
290 IF C=0 THEN 1270
300 IF C<>-7 THEN PX=XP:PY=YP
310 LOCATE XP,YP
320 PRINT N$;
330 LOCATE PX,PY
340 COLOR 1
350 PRINT P$;
360 YP=PY
370 XP=PX
380 VX=VX+CX
390 VY=VY+CY
400 IF SCREEN(VX,VY)<>32 THEN GOSUB 660
410 IF SCREEN(VX,VY)<>32 THEN 380
420 LOCATE XV,YV
430 COLOR 0
440 PRINT N$;
450 LOCATE VX,VY
460 PRINT V$;
470 XV=VX
480 YV=VY
490 GOTO 110
500 IF INKEY$<>" " THEN 500
```

```
510 IF R<S THEN R=S
520 COLOR 0
530 LOCATE 10,6
540 PRINT "TEMPS ECOULE";
550 LOCATE 10,10
560 PRINT "SCORE :";S;
570 LOCATE 10,14
580 PRINT "RECORD :";R;
590 LOCATE 10,18
600 PRINT "UNE AUTRE ?";
610 D$=INKEY$
620 IF D$="" THEN 610
630 IF D$<>"N" THEN RUN
640 CLS
650 END
660 DT=DT+1
670 GOSUB 770
680 IF SCREEN(XV+CX,YV+CY)=32 THEN VX=XV
+CX:VY=YV+CY:RETURN
690 DT=DT-2
700 GOSUB 770
710 IF SCREEN(XV+CX,YV+CY)=32 THEN VX=XV
+CX:VY=YV+CY:RETURN
720 DT=DT-1
730 GOSUB 770
740 VX=XV+CX
750 VY=YV+CY
760 RETURN
770 IF DT>4 THEN DT=DT-4
780 IF DT<1 THEN DT=DT+4
790 CX=(DT=1)-(DT=3)
800 CY=(DT=2)-(DT=4)
810 RETURN
820 CLS
830 COLOR 5
840 FOR VX=0 TO 39
850 LOCATE VX,0
860 PRINT CHR$(127);
870 LOCATE VX,23
880 PRINT CHR$(127);
890 NEXT VX
900 FOR VY=1 TO 22
910 LOCATE 0,VY
```

```
920 PRINT CHR$(127);
930 LOCATE 39,VY
940 PRINT CHR$(127);
950 NEXT VY
960 COLOR 4
970 FOR VX=1 TO 150
980 GOSUB 1230
990 LOCATE PX,PY
1000 PRINT CHR$(127);
1010 NEXT VX
1020 GOSUB 1230
1030 VX=PX
1040 VY=PY
1050 COLOR 0
1060 LOCATE VX,VY
1070 PRINT V$;
1080 XV=VX
1090 YV=VY
1100 GOSUB 1230
1110 COLOR 1
1120 LOCATE PX,PY
1130 PRINT P$;
1140 XP=PX
1150 YP=PY
1160 Z=30
1170 CX=0
1180 CY=0
1190 DX=0
1200 DY=0
1210 DT=0
1220 RETURN
1230 PX=INT(RND*38)+1
1240 PY=INT(RND*22)+1
1250 IF SCREEN(PX,PY)<>32 THEN 1230
1260 RETURN
1270 FOR I=1 TO 5
1280 PLAY "T15REDD"
1290 NEXT I
1300 S=S+1
1310 GOTO 100
1320 SCREEN 4,6,6
1330 CLS
1340 DEFINT A-Y
```

```
1350 DEFGR$(0)=28,28,200,62,9,28,20,20
1360 DEFGR$(1)=0,126,255,153,255,126,60,
0
1370 ATTRB 1,1
1380 LOCATE 10,10,0
1390 PRINT "JOYSTICK ?";
1400 D$=INKEY$
1410 PX=RND
1420 IF D$="" THEN 1400
1430 IF D$="0" THEN JK=1
1440 ATTRB 0,0
1450 RETURN
```

ROBOTS

Vous voici seul, abandonné sur une planète défendue par des robots meurtriers. Le sol est truffé de mines qu'il vous faut absolument éviter. Les mines sont représentées sur l'écran par des losanges rouges. Au début du jeu, cinq robots sont présents sur le terrain. Sans perdre une seconde, ils se précipitent sur vous en suivant toujours le plus court chemin. Par chance, les robots sont aveugles et ne voient pas les mines qui se trouvent entre eux et vous, ce qui vous permet, en vous déplaçant judicieusement, de les éliminer. Utilisez pour cela le joystick ou les touches :

< A >	< Z >	< E >
< Q >		< D >
< W >	< X >	< C >

suivant la direction que vous avez choisie. Lorsque tous les robots sont éliminés, le jeu reprend avec un robot supplémentaire. Si vous sautez sur une mine ou si un robot vous tue, tout n'est pas perdu. Vous disposez en effet de cinq vies. Si vous désirez changer le nombre de mines, modifiez la valeur de la variable NM à la ligne 80.


```
10 REM *****
20 REM * ROBOTS *
30 REM *****
40 DEFINT A-Z
50 CLEAR , , 3
60 NH=5
70 N1=5
```

```
80 NM=40
90 NR=N1
100 DIM R(30,1)
110 GOSUB 1580
120 GOSUB 1470
130 GOSUB 910
140 ON JS GOSUB 710,810
150 C=POINT(HX*8+4, HY*8+4)
160 IF C<>-3 AND C<>4 THEN 470
170 COLOR 4
180 LOCATE X,Y
190 PRINT N$;
200 LOCATE HX, HY
210 PRINT H$;
220 X=HX
230 Y=HY
240 T=0
250 FOR I=1 TO NR
260 IF R(I,0)=0 THEN 400
270 T=1
280 RX=R(I,0)+SGN(HX-R(I,0))
290 RY=R(I,1)+SGN(HY-R(I,1))
300 C=POINT(RX*8+4, RY*8+4)
310 IF C=1 OR C=0 THEN S=S+1: LOCATE R(I,
0), R(I,1): PRINT N$;: R(I,0)=0: GOTO 400
320 IF C=4 THEN 470
330 COLOR 0
340 LOCATE R(I,0), R(I,1)
350 PRINT N$;
360 LOCATE RX, RY
370 PRINT R$;
380 R(I,0)=RX
390 R(I,1)=RY
400 NEXT I
410 IF T=0 THEN 430
420 GOTO 140
430 S=S+10
440 IF INKEY$<>" " THEN 440
450 IF NR<30 THEN NR=NR+1
460 GOTO 130
470 NH=NH-1
480 COLOR 7
490 LOCATE X,Y
```

```
500 PRINT N$;
510 LOCATE HX,HY
520 PRINT H$;
530 PLAY "L96REL72REL24REL96REL72FAL24MI
L72MIL24REL72REL24DO#L96RE"
540 IF INKEY$<>"" THEN 540
550 IF NH>0 THEN NR=N1:GOTO 130
560 CLS
570 SCREEN 1,6,6
580 ATTRB 1,1
590 LOCATE 9,10
600 PRINT "SCORE :";S;
610 LOCATE 9,20
620 PRINT "UNE AUTRE ?";
630 COLOR 4
640 ATTRB 0,0
650 IF INKEY$<>"" THEN 650
660 D$=INKEY$
670 IF D$="" THEN 660
680 IF D$<>"N" THEN RUN
690 CLS
700 END
710 D$=INKEY$
720 IF D$="A" THEN HX=HX-1:HY=HY-1
730 IF D$="Z" THEN HY=HY-1
740 IF D$="E" THEN HY=HY-1:HX=HX+1
750 IF D$="Q" THEN HX=HX-1
760 IF D$="D" THEN HX=HX+1
770 IF D$="W" THEN HX=HX-1:HY=HY+1
780 IF D$="X" THEN HY=HY+1
790 IF D$="C" THEN HY=HY+1:HX=HX+1
800 RETURN
810 J=STICK(0)
820 IF J=1 THEN HY=HY-1
830 IF J=2 THEN HY=HY-1:HX=HX+1
840 IF J=3 THEN HX=HX+1
850 IF J=4 THEN HX=HX+1:HY=HY+1
860 IF J=5 THEN HY=HY+1
870 IF J=6 THEN HY=HY+1:HX=HX-1
880 IF J=7 THEN HX=HX-1
890 IF J=8 THEN HX=HX-1:HY=HY-1
900 RETURN
910 CLS
```

```
920 COLOR 4
930 LOCATE 0,24
940 PRINT "SCORE :";S;
950 IF NH=1 THEN 1000
960 FOR HX=1 TO NH-1
970 LOCATE 19+HX,24
980 PRINT H$;
990 NEXT HX
1000 COLOR 1
1010 FOR HX=0 TO 39
1020 LOCATE HX,0
1030 PRINT M$;
1040 LOCATE HX,23
1050 PRINT M$;
1060 NEXT HX
1070 FOR HY=1 TO 22
1080 LOCATE 0,HY
1090 PRINT M$;
1100 LOCATE 39,HY
1110 PRINT M$;
1120 NEXT HY
1130 FOR I=1 TO NM
1140 HX=INT(RND*38)+1
1150 HY=INT(RND*22)+1
1160 IF SCREEN(HX,HY)<>32 THEN 1140
1170 LOCATE HX,HY
1180 PRINT M$;
1190 NEXT I
1200 COLOR 0
1210 FOR I=1 TO NR
1220 R(I,0)=INT(RND*38)+1
1230 R(I,1)=INT(RND*22)+1
1240 IF SCREEN(R(I,0),R(I,1))<>32 THEN 1
220
1250 LOCATE R(I,0),R(I,1)
1260 PRINT R$;
1270 NEXT I
1280 HX=INT(RND*38)+1
1290 HY=INT(RND*22)+1
1300 IF SCREEN(HX,HY)<>32 THEN 1280
1310 X=HX
1320 Y=HY
1330 FOR I=1 TO 5
```

```
1340 LOCATE HX, HY
1350 COLOR 5
1360 PRINT CHR$(127);
1370 BEEP
1380 FOR J=1 TO 50
1390 NEXT J
1400 LOCATE HX, HY
1410 COLOR 4
1420 PRINT H$;
1430 FOR J=1 TO 50
1440 NEXT J
1450 NEXT I
1460 RETURN
1470 CLS
1480 SCREEN 4, 2, 0
1490 ATTRB 1, 1
1500 LOCATE 10, 10, 0
1510 PRINT "JOYSTICK ?";
1520 ATTRB 0, 0
1530 D$=INKEY$
1540 C=RND
1550 IF D$="" THEN 1530
1560 IF D$="0" THEN JS=2 ELSE JS=1
1570 RETURN
1580 DEFGR$(0)=28, 28, 73, 62, 8, 28, 20, 20
1590 DEFGR$(1)=60, 126, 219, 255, 255, 126, 36
, 60
1600 DEFGR$(2)=0, 0, 24, 60, 126, 126, 60, 24
1610 H$=GR$(0)
1620 R$=GR$(1)
1630 M$=GR$(2)
1640 N$=CHR$(32)
1650 RETURN
```



PARACHUTE

Essayez, en sautant d'un hélicoptère en vol, d'atteindre la cible qui se trouve au sol. Une première pression sur une touche vous permet de descendre verticalement en chute libre. Une deuxième pression entraîne l'ouverture du parachute. La descente continue alors plus lentement et avec un angle de 45 degrés car le vent vous pousse. Plus vous attendrez pour ouvrir votre parachute et moins vous serez déporté. Mais n'attendez pas trop longtemps car, au-dessous de cent mètres, le parachute ne s'ouvre plus.

```
10 REM *****
20 REM * PARACHUTE *
30 REM *****
40 GOSUB 470
50 HH=HH-2
60 IF HH=0 THEN PUT (0,1)-(25,9),R
70 IF HH=0 THEN HH=290
80 PUT (HH,1)-(HH+25,9),H
90 D$=INKEY$
100 IF D$="" THEN 140
110 IF PV>100 THEN 140
120 IF SP=1 THEN OP=1 ELSE SP=1
130 IF OP=0 THEN PV=10:PH=HH
140 IF SP=0 THEN 230
150 IF OP=0 THEN PV=PV+4
160 IF OP=1 THEN PV=PV+1:PH=PH-1
170 IF PV>167 OR PH<1 THEN 260
180 IF OP=1 THEN 210
190 PUT (PH,PV)-(PH+14,PV+23),PF
200 GOTO 50
```



```
210 PUT (PH,PV)-(PH+14,PV+23),PO
220 GOTO 50
230 FOR I=1 TO 50
240 NEXT I
250 GOTO 50
260 IF ABS(PH-A)>4 THEN 320
270 FOR I=1 TO 1000
280 NEXT I
290 S=S+10
300 GOSUB 670
310 GOTO 50
320 CLS
330 SCREEN 1,2,2
340 LOCATE 10,10
350 ATTRB 1,1
360 PRINT "SCORE :";S;
370 LOCATE 10,16
380 PRINT "UNE AUTRE ?";
390 ATTRB 0,0
400 IF INKEY$<>" " THEN 400
410 D$=INKEY$
420 IF D$="" THEN 410
430 IF D$<>"N" THEN RUN
440 CLS
450 SCREEN 4,6,6
460 END
470 LOCATE 0,0,0
480 SCREEN 4,6,6
490 DEFINT A-Z
500 DIM H(25,8)
510 DIM PF(14,23)
520 DIM PO(14,23)
530 DIM R(25,8)
540 CLS
550 DRAW "C4BM51,50R14L6G4L2G1L1G1D1F1R2
1E1U5L2D1G2L5H1L1H1L1H1L1"
560 PAINT (58,56),4
570 GET (50,50)-(75,58),H
580 GET (100,50)-(125,58),R
590 CLS
600 DRAW "C1BM54,50G1L1G1D1G1D1R12U1H1U1
H1L1H1L4"
610 PAINT (56,54),1
```

```
620 DRAW "C1BM51,56D1F1D1F1D3R2D6L1R1U3R
2D3R1L1U4L1U4L1U1R2D1L1D4R1U2R2U3E1U1E1U
1"
630 GET (50,46)-(64,69),PO
640 CLS
650 DRAW "C1BM54,61R1D6L2D2U2R3D2L1D1R2U
1L1U2R3D2U2L3U3D3R1U6R1"
660 GET (50,46)-(64,69),PF
670 CLS
680 HH=290
690 HV=1
700 A=INT(RND*191)+10
710 DRAW "COBM"+STR$(A)+"",191R12"
720 SP=0
730 OP=0
740 PV=0
750 PH=0
760 RETURN
```

LA BIBLIOTHÈQUE SYBEX

OUVRAGES GÉNÉRAUX

VOTRE PREMIER ORDINATEUR *par RODNAY ZAKS*,
296 pages, Réf. 226

VOTRE ORDINATEUR ET VOUS *par RODNAY ZAKS*,
296 pages, Réf. 271

DU COMPOSANT AU SYSTÈME : une introduction aux microprocesseurs *par RODNAY ZAKS*,
636 pages, Réf. 239

TECHNIQUES D'INTERFACE aux microprocesseurs *par AUSTIN LESEA ET RODNAY ZAKS*,
450 pages, Réf. 230, 3ème édition

LEXIQUE INTERNATIONAL MICROORDINATEURS, avec dictionnaire abrégé en 10 langues
192 pages, Réf. 234

BASIC

VOTRE PREMIER PROGRAMME BASIC *par RODNAY ZAKS*,
208 pages, Réf. 263

INTRODUCTION AU BASIC *par PIERRE LE BEUX*,
336 pages, Réf. 216

LE BASIC PAR LA PRATIQUE : 60 exercices *par JEAN-PIERRE LAMOITIER*,
252 pages, Réf. 231

LE BASIC POUR L'ENTREPRISE *par XUAN TUNG BUI*,
204 pages, Réf. 253, 2ème édition

PROGRAMMES EN BASIC, Mathématiques, Statistiques, Informatique *par ALAN R. MILLER*,
318 pages, Réf. 259

AU COEUR DES JEUX EN BASIC *par RICHARD MATEOSIAN*,
352 pages, Réf. 233

JEUX D'ORDINATEUR EN BASIC *par DAVID H. AHL*,
192 pages, Réf. 246

NOUVEAUX JEUX D'ORDINATEUR EN BASIC *par DAVID H. AHL*,
204 pages, Réf. 247

PASCAL

INTRODUCTION AU PASCAL *par PIERRE LE BEUX*,
496 pages, Réf. 222

LE PASCAL PAR LA PRATIQUE *par PIERRE LE BEUX ET HENRI TAVERNIER,*
562 pages, Réf. 229

LE GUIDE DU PASCAL *par JACQUES TIBERGHEN,*
504 pages, Réf. 232

PROGRAMMES EN PASCAL pour Scientifiques et Ingénieurs *par ALAN R. MILLER,*
392 pages, Réf. 240

AUTRES LANGAGES

INTRODUCTION A ADA *par PIERRE LE BEUX,*
366 pages, Réf. 242

MICROORDINATEUR

ALICE

JEUX EN BASIC POUR ALICE *par PIERRE MONSAUT,*
96 pages, Réf. 320

APPLE

PROGRAMMEZ EN BASIC SUR APPLE II *par LÉOPOLD LAURENT,*
208 pages, tome 1, Réf. 268

APPLE II 66 PROGRAMMES BASIC *par STANLEY R. TROST,*
192 pages, Réf. 283

JEUX EN PASCAL SUR APPLE *par DOUGLAS HERGERT ET JOSEPH T. KALASH,*
372 pages, Réf. 241

ATARI

JEUX EN BASIC SUR ATARI *par PAUL BUNN,*
96 pages, Réf. 282

COMMODORE 64

JEUX EN BASIC SUR COMMODORE 64 *par PIERRE MONSAUT,*
96 pages, Réf. 317

DRAGON

JEUX EN BASIC SUR DRAGON *par PIERRE MONSAUT,*
96 pages, Réf. 324

GOUPIL

PROGRAMMEZ VOS JEUX SUR GOUPIL *par FRANÇOIS ABELLA,*
208 pages, Réf. 264

IBM

EXERCICES EN BASIC SUR L'ORDINATEUR PERSONNEL IBM *par JEAN-PIERRE LAMOITIER*,
256 pages, Réf. 267

IBM PC Guide de l'utilisateur, *par JOAN LASSELLE ET CAROL RAMSEY*,
160 pages, Réf. 301

IBM PC 66 PROGRAMMES BASIC *par STANLEY R. TROST*,
192 pages, Réf. 280

ORIC

JEUX EN BASIC SUR ORIC *par PETER SHAW*,
96 pages, Réf. 278

SHARP

DÉCOUVREZ LE SHARP PC-1500 ET LE TRS-80 PC-2 *par MICHEL LHOIR*,
2 tomes, Réf. 261-262

SPECTRUM

PROGRAMMEZ EN BASIC SUR SPECTRUM *par S.M. GEE*,
208 pages, Réf. 252

JEUX EN BASIC SUR SPECTRUM *par PETER SHAW*,
96 pages, Réf. 276

TIMEX

DÉCOUVREZ LE ZX 81 ET LE TIMEX SINCLAIR 1000 *par DOUGLAS HERGERT*,
208 pages, Réf. 256

TRS-80

PROGRAMMEZ EN BASIC SUR TRS-80 *par LÉOPOLD LAURENT*,
2 tomes, Réf. 250-251

DÉCOUVREZ LE SHARP PC-1500 ET LE TRS-80 PC-2 *par MICHEL LHOIR*,
2 tomes, Réf. 261-262

JEUX EN BASIC SUR TRS-80 MC-10 *par PIERRE MONSAUT*,
96 pages, Réf. 323

JEUX EN BASIC SUR TRS-80 *par CHRIS PALMER*,
96 pages, Réf. 302

JEUX EN BASIC SUR TRS-80 COULEUR *par PIERRE MONSAUT*,
96 pages, Réf. 325

VIC 20

PROGRAMMEZ EN BASIC SUR VIC 20 *par G. O. HAMANN*,
2 tomes, Réf. 244-245

JEUX EN BASIC SUR VIC 20 *par ALASTAIR GOURLAY,*
96 pages, Réf. 277

ZX 81

DÉCOUVREZ LE ZX 81 ET LE TIMEX SINCLAIR 1000 *par DOUGLAS HERGERT,*
208 pages, Réf. 256

ZX 81 56 PROGRAMMES BASIC *par STANLEY R. TROST,*
192 pages, Réf. 304

GUIDE DU BASIC ZX 81 *par DOUGLAS HERGERT,*
204 pages, Réf. 285

JEUX EN BASIC SUR ZX 81 *par MARK CHARLTON,*
96 pages, Réf. 275

MICRO-PROCESSEURS

PROGRAMMATION DU Z80 *par RODNAY ZAKS,*
618 pages, Réf. 220

APPLICATIONS DU Z80 *par JAMES W. COFFRON,*
304 pages, Réf. 274

PROGRAMMATION DU 6502 *par RODNAY ZAKS,*
376 pages, Réf. 224, 2ème édition

APPLICATIONS DU 6502 *par RODNAY ZAKS,*
288 pages, Réf. 219

PROGRAMMATION DU 6800 *par DANIEL-JEAN DAVID ET RODNAY ZAKS,*
374 pages, Réf. 218

PROGRAMMATION DU 6809 *par RODNAY ZAKS ET WILLIAM LABIAK,*
392 pages, Réf. 270

SYSTÈMES D'EXPLOITATION

GUIDE DU CP/M AVEC MP/M *par RODNAY ZAKS,*
354 pages, Réf. 228

CP/M APPROFONDI *par ALAN R. MILLER,*
380 pages, Réf. 265

INTRODUCTION AU p-SYSTEM UCSD *par CHARLES W. GRANT ET JON BUTAH,*
308 pages, Réf. 257

LOGICIELS ET APPLICATIONS

INTRODUCTION AU TRAITEMENT DE TEXTE *par HAL GLATZER,*
228 pages, Réf. 243

INTRODUCTION A WORDSTAR *par ARTHUR NAIMAN,*
200 pages, Réf. 255

VISICALC APPLICATIONS *par STANLEY R. TROST,*
304 pages, Réf. 258

La plupart de ces ouvrages existent en version anglaise. N'hésitez pas à demander notre catalogue.

EN ANGLAIS

BASIC EXERCISES FOR APPLE *by JEAN-PIERRE LAMOITIER,*
232 pages, Réf. 0-084

BASIC FOR BUSINESS *by DOUGLAS HERGERT,*
224 pages, Réf. 0-080

CELESTIAL BASIC : Astronomy on your Computer *by ERIC BURGESS,*
228 pages, Réf. 0-087

INTRODUCTION TO PASCAL (Including UCSD Pascal) *by RODNAY ZAKS,*
422 pages, Réf. 0-066

DOING BUSINESS WITH PASCAL *by RICHARD HERGERT AND DOUGLAS HERGERT,*
380 pages, Réf. 0-091

MASTERING VISICALC *by DOUGLAS HERGERT,*
224 pages, Réf. 0-090

THE APPLE CONNECTION *by JAMES W. COFFRON,*
228 pages, Réf. 0-085

PROGRAMMING THE Z8000 *by RICHARD MATEOSIAN,*
300 pages, Réf. 0-032

A MICROPROGRAMMED APL IMPLEMENTATION *by RODNAY ZAKS,*
350 pages, Réf. 0-005

ADVANCED 6502 PROGRAMMING *by RODNAY ZAKS,*
292 pages, Réf. 0-089

FORTTRAN PROGRAMS FOR SCIENTISTS AND ENGINEERS *by ALAN R. MILLER,*
320 pages, Réf. 0-082